CHAPTER 6

行列と 1 次変換・ParametricPlot

ベクトルや行列がリスト（のリスト）の形で表現されることは 3 章でみた．ここではリストのベクトル・行列としての演算をまとめておく．

また，組込み関数 ParametricPlot と ParametricPlot3D を駆使して，行列が定める線形写像の作用を可視化する方法を紹介しよう．

6.1　行列とベクトルの演算

[1] **ベクトル**を定義[*1]：

```
In[1]:= v1 = {x, y}; v2 = {z, w};
```

[2] **行列**を定義：

```
In[2]:= m1 = {{a, b}, {c, d}};
        m2 = {{3, 2}, {1, 4}};
```

[3] 行列として表示（3 章 3.3 節）：

```
In[3]:= m1 //MatrixForm
```

Out[3]//MatrixForm= $\begin{pmatrix} a & b \\ c & d \end{pmatrix}$

[4] ベクトルの**内積**：

```
In[4]:= v1.v2
```

Out[4]= w y + x z

*1 v1//MatrixForm とすると，縦ベクトル $\begin{pmatrix} x \\ y \end{pmatrix}$ （2 行 1 列の行列）であることがわかる．

[5] ベクトルと行列の積：

```
In[5]:= m1.v1
Out[5]= {a x + b y, c x + d y}
```

[6] 行列と行列の積（pr とおく）：

```
In[6]:= pr = m1.m2
Out[6]= {{3 a + b, 2 a + 4 b}, {3 c + d, 2 c + 4 d}}
```

[7] 行列形で表現：

```
In[7]:= pr //MatrixForm
```

Out[7]//MatrixForm= $\begin{pmatrix} 3\,a+b & 2\,a+4\,b \\ 3\,c+d & 2\,c+4\,d \end{pmatrix}$

[8] MatrixPower で行列の**べき乗** (power) を計算できる．たとえば m2 の 7 乗：

```
In[8]:= MatrixPower[m2, 7] //MatrixForm
```

Out[8]//MatrixForm= $\begin{pmatrix} 26\,127 & 51\,998 \\ 25\,999 & 52\,126 \end{pmatrix}$

[9] Transpose で m1 の**転置行列** (transposed matrix) を求める：

```
In[9]:= Transpose[m1] //MatrixForm
```

Out[9]//MatrixForm= $\begin{pmatrix} a & c \\ b & d \end{pmatrix}$

[10] Inverse で**逆行列** (inverse matrix) を求める：

```
In[10]:= Inverse[m1] //MatrixForm
```

Out[10]//MatrixForm= $\begin{pmatrix} \frac{d}{-b\,c+a\,d} & -\frac{b}{-b\,c+a\,d} \\ -\frac{c}{-b\,c+a\,d} & \frac{a}{-b\,c+a\,d} \end{pmatrix}$

[11] Det で**行列式** (determinant) を求める：

```
In[11]:= Det[m1]
Out[11]= -b c + a d
```

[12] Tr で**トレース** (trace) を求める：

```
In[12]:= Tr[m1]
Out[12]= a + d
```

[13] 2 次の**単位行列** (identity matrix) を i2 とおく：

```
In[13]:= i2 = IdentityMatrix[2]
```

Out[13]=　{{1, 0}, {0, 1}}

同様に n 次の単位行列は `IdentityMatrix[n]` で得られる．

問題 6.1（ケーリー・ハミルトンの等式） [6][11][12][13] を用いて，任意の 2 次正方行列 $M = \begin{pmatrix} a & b \\ c & d \end{pmatrix}$ は等式 $M^2 - \mathrm{tr}(M)M + \det(M)I_2 = O$ をみたすことを示せ．ただし I_2 は 2 次の単位行列，O はゼロ行列とする．

【解答】 $M = \begin{pmatrix} a & b \\ c & d \end{pmatrix}$ には [2] の `m1` を，I_2 には [13] の `i2` を流用しよう．$M^2 - \mathrm{tr}(M)M + \det(M)I_2$ に対応する式を入力し，`Simplify` を施してみる：

```
In[ ]:=  Simplify[m1.m1 - Tr[m1]*m1 + Det[m1]*i2]

Out[ ]=  {{0, 0}, {0, 0}}
```

これはゼロ行列 O である．■

6.2　固有値・固有ベクトル

n 次正方行列 M に対し，複素数 λ と n 次元ベクトル v が $Mv = \lambda v$ をみたすとき，λ を M の**固有値** (eigenvalue)，v を（λ に関する）**固有ベクトル** (eigenvector) とよぶ．*Mathematica* にこれらを計算させるには，組込み関数 `Eigensystem` を用いるとよい．

[14] [2] で定義した行列 $\begin{pmatrix} 3 & 2 \\ 1 & 4 \end{pmatrix}$ を `m` とおく：

```
In[14]:=  m = m2;
```

[15] `Eigensystem` で **固有値・固有ベクトル**を求める（`es` とおく）：

```
In[15]:=  es = Eigensystem[m]

Out[15]=  {{5, 2}, {{1, 1}, {-2, 1}}}
```

[16] 一見わかりづらいが，`MatrixForm` でみると，対応する固有値・固有ベクトルが縦に並んでいることがわかる：

```
In[16]:=  es // MatrixForm

Out[16]//MatrixForm=
   (    5         2     )
   ( {1, 1}   {-2, 1}   )
```

Out[ ]=



■

問題 6.4 (単位円板の像) 関数 $\mathrm{disk}(r,t) = (r\cos t, r\sin t)$ （ただし $0 \leq r \leq 1, 0 \leq t \leq 2\pi$）を定義し，単位円板とその f による像を ParametricPlot で図示せよ．

【解答】 [30][31][32] を参考にして，次のように入力する：

```
In[ ]:=  disk[r_, t_] = {r Cos[t], r Sin[t]};

In[ ]:=  ParametricPlot[{disk[r, t], f[disk[r, t]]},
                                   {r, 0, 1}, {t, 0, 2 Pi}]
```

Out[ ]=



■

6.4　3 次元の 1 次変換と ParametricPlot3D

xyz 空間 $\mathbb{R}^3$ に対し，1 次変換 $g : \mathbb{R}^3 \to \mathbb{R}^3$ を

$$g\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} 1 & 0 & 2 \\ 0 & 2 & 0 \\ 2 & 0 & 1 \end{pmatrix}\begin{pmatrix} x \\ y \\ z \end{pmatrix} = \begin{pmatrix} x+2z \\ 2y \\ 2x+z \end{pmatrix}$$

で定義する．今度は組込み関数 ParametricPlot3D を用いてその作用を可視化しよう．考え方は平面のときとまったく同じである．

[33] 行列を定義：

In[33]:=
```
mm = {{1, 0, 2}, {0, 2, 0}, {2, 0, 1}};
mm // MatrixForm
```

Out[33]//MatrixForm=
$$\begin{pmatrix} 1 & 0 & 2 \\ 0 & 2 & 0 \\ 2 & 0 & 1 \end{pmatrix}$$

[34] 図示とは直接関係ないが，固有値・ベクトルを確認しておく：

In[34]:=
```
Eigensystem[mm] // MatrixForm
```

Out[34]//MatrixForm=
$$\begin{pmatrix} 3 & 2 & -1 \\ \{1,\ 0,\ 1\} & \{0,\ 1,\ 0\} & \{-1,\ 0,\ 1\} \end{pmatrix}$$

[35] 1 次変換を定義：

In[35]:=
```
g[{x_, y_, z_}] = mm.{x, y, z}
```

Out[35]= {x + 2 z, 2 y, 2 x + z}

[36] 曲線 $\{(\cos t,\ \sin t,\ t/10) \mid 0 \le t \le 10\pi\}$ を考えよう．これは**螺旋** (spiral もしくは helix) の一部である．これと g による像を図示してみよう．まず螺旋を与える空間ベクトル値関数を定義：

In[36]:=
```
spi[t_] = {Cos[t], Sin[t], t/10};
```

[37] ParametricPlot3D でその 3 次元グラフを描く：

In[37]:=
```
ParametricPlot3D[spi[t], {t, 0, 10 Pi}]
```

Out[37]=


出力部分をマウスでドラッグすると回転などの操作ができる（4 章 4.5 節）．

[38] g による像：

In[38]:=
```
ParametricPlot3D[g[spi[t]], {t, 0, 10 Pi}]
```

Out[38]=



[39] 同一座標内で比較する：

```
In[39]:= ParametricPlot3D[{spi[t], g[spi[t]]}, {t, 0, 10 Pi}]
```

Out[39]=



[40] ParametricPlot3D も 2 つのパラメーターに対応している．単位球面（sphere）

$$\{(\cos s \sin t,\ \ \sin s \sin t,\ \ \cos t) \mid 0 \leq s \leq 2\pi,\ 0 \leq t \leq \pi\}$$

とその像を描かせてみよう（いわば問題 6.4 の 3 次元版である）：

```
In[40]:= sph[s_, t_] = {Cos[s] Sin[t], Sin[s] Sin[t], Cos[t]};
         ParametricPlot3D[ {sph[s, t], g[sph[s, t]]},
                   {s, 0, 2 Pi}, {t, 0, Pi},
                   PlotStyle ->
                   {Opacity[0.9, Green], Opacity[0.4, Yellow]}]
```

Out[40]=



オプションでは，2 つの曲面 sph[s, t]，g[sph[s, t]] それぞれに別々の PlotStyle を指定している．Opacity とは**不透明度**のことで，sph[s, t] は不透明度 0.9 で緑色，g[sph[s, t]] は不透明度 0.4 で黄色に設定した．

問題 6.5 (トーラス) 以下で定義される**トーラス** (torus，円環面) について，そのグラフと g による像のグラフを描け．また，それらを同一空間に描画せよ．

$$\{((3+\cos t)\cos s,\ (3+\cos t)\sin s,\ \sin t) \mid 0 \leq s \leq 2\pi,\ 0 \leq t \leq 2\pi\}$$

【解答】 トーラスのパラメーター表示をたとえば

```
tor[s_, t_] = { Cos[s] (3 + Cos[t]),
                Sin[s] (3 + Cos[t]),
                Sin[t] }
```

のように定義し t の変域を {t, 0, 2 Pi} とすれば，あとは [40] と同じである．右の絵は透明度などのオプションを指定せずに描いた．



■

6.5 研究：1 次変換による「向きの変化」を表現する

1 次変換には「向きを保つ」ものと「向きを変える」ものがある．それは 1 次変換を定義する行列の行列式の値が正か負かに対応している．

たとえば上で定義した 1 次変換 g の場合，mm で Det[mm] を計算すると -6 となり「向きを変える」変換であることがわかる．一体どういうことなのか，その様子を観察してみよう．

[41] xy, yz, zx 平面に対応する関数を定義：

In[41]:=
```
xy[s_, t_] = {s, t, 0};
```

In[41]:=
```
yz[s_, t_] = {0, s, t};
```

In[41]:=
```
zx[s_, t_] = {t, 0, s};
```

[42] xy, yz, zx 平面を部分的に切り取った 3 枚の板 (panels) を描く：

In[42]:=
```
pan = ParametricPlot3D[
        {xy[s, t], yz[s, t], zx[s, t]},
        {s, -1, 3}, {t, -1, 3},
        PlotStyle -> {Red, Green, Blue}]
```

Out[42]=



s，t の範囲を原点対称にしないのがポイントで，これにより x, y, z 軸の「正の方向」が示唆される．

[43] 板の g による像を描く：

In[43]:=
```
pan2 = ParametricPlot3D[
        {g[xy[s, t]], g[yz[s, t]], g[zx[s, t]]},
        {s, -1, 3}, {t, -1, 3},
        PlotStyle -> {Red, Green, Blue}]
```

Out[43]=



この結果と [42] を見比べると，3 枚の板の位置関係が変化していることがわかる．g は板たちをただ斜めに引き伸ばすのではなく，いったん鏡像をとってから引き伸ばし

ているのである．「向きを変える・保つ」という属性はこの鏡像をとる操作が入るか入らないかに対応するのである．

[44] 両方の絵をまとめて表示：

```
In[44]:= Show[pan2, pan, Boxed -> False, Axes -> False]
```

Out[44]=



Boxed -> False , Axes -> False はそれぞれ枠線と軸を消すためのオプションである．ちなみに Show[pan, pan2, ...] という順序にすると，描画範囲が前者 pan に調整されて，pan2 が部分的にしか描画されない．

6.6　研究

(1) ドキュメントセンターで行列の階数 (rank) を計算する MatrixRank，行列の指数関数を計算する MatrixExp，行列のジョルダン標準形を求める JordanDecomposition について調べ，使い方を確認せよ．

(2) ParametricPlot3D を用いて，土星（っぽいもの）を描け．（Hint．オプション Mesh -> None を用いるとグラフの網目が消えてそれらしくなる．）

(3) ParametricPlot3D を用いて**メビウスの帯**を描け．（Hint．下の図をよく眺めると，円周にそって線分を回転させながら移動させた軌跡に見えるはず．）